
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2020-09_D22wk23 related to the title of this CR.

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc109680266]Corrections required

0. [bookmark: _Toc106975634][bookmark: _Toc109680267]Change 1
	Function name
	f_IMS_OtherResponse_200_MessageHeaderRX()

	Reason for change
	Currently if this function is called for condition A5 (response sent within a dialog), the contact URI to expect is retrieved from the stored dialog information (f_IMS_PTC_ImsInfo_GetContactUrl()). This can cause a problem as the stored URI can have URL parameters like “transport=” which can vary across transactions within a dialog. For example if a UE sends the initial INVITE establishing a dialog using TCP transport, and the SS later sends a re-INVITE using UDP, then the checking of the contact header in the following 200 OK will fail as it will expect url parameter “transport=tcp” in the contact header but UE will include “transport=udp”.

	Summary of change
	Allow any url parameters in contact header (host address and port will still be checked as per stored contact information for the dialog).

	TTCN module
	IMS_Procedures_Common.ttcn

	MCC160 Comment
	



Before Change
	function f_IMS_OtherResponse_200_MessageHeaderRX(template (value) MessageHeader p_MessageHeader_Request,
                                                   template ContentType p_ContentType := omit,
                                                   boolean p_PreconditionRequiredForSDP := true) runs on IMS_PTC return template (present) MessageHeader
  { /* NOTE: Message is sent by the UE => conditions A2, A4 and A5 are applicable;
     *       For A4 (GIBA) pAccessNetworkInfo is optional */
    /* @sic R5-130683 sic@ */
    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */
    /* @sic R5-160790, R5-160903, R5-160793: p_PreconditionIndication sic@ */
    /* @sic R5-162930: p_PreconditionIndication := PreconditionRequired => per default there is a Require header if the message contains an SDP sic@ */
    /* @sic R5-174544: SessionId removed sic@ */
    /* @sic R5s210615: p_PreconditionsRequiredForSDP (boolean) instead of p_PreconditionIndication (IMS_PreconditionIndication_Type) sic@ */
    var template (present) MessageHeader v_MessageHeader_Response := f_IMS_BaseResponse_MessageHeaderRX(p_MessageHeader_Request);     /* @sic R5s200000: f_IMS_BaseResponse_MessageHeaderRX sic@ */
    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());  /* acc. to A.3.1 condition A5: access network technology and, if applicable, the cell ID */
    var boolean v_A2A4 := false;                                        /* true => response sent by UE for INVITE/UPDATE */
    var boolean v_A5 := f_IMS_PTC_ImsInfo_DialogIsEstablished();        /* true => response sent within a dialog */
    var boolean v_A8 := true;                                           /* true => any response but not CANCEL */
    var boolean v_A20 := false;                                         /* true => response sent by UE for INVITE */
    var boolean v_A18 := pc_IMS_Video_FeatureTag;                       /* true => UE supports video feature tag */
    var boolean v_A23 := false;                                         /* true => response sent for re-INVITE within an established dialog */
    var charstring v_Method := valueof(p_MessageHeader_Request.cSeq.method);
    var integer v_Port_us := f_IMS_PTC_GetPort_us();  /* @sic R5s130985 Change 2 sic@ */
    var template (present) SemicolonParam_List v_MandatoryContactParams := {};
    var template (present) SipUrl v_ContactURI;

    select (v_Method) {
      case ("UPDATE") {
        v_A2A4 := true;
      }
      case ("INVITE") {
        v_A2A4 := true;
        v_A20 := true;
        v_A23 := v_A5;
      }
      case ("CANCEL") {
        v_A8 := false;
      }
    }
    
    if (p_PreconditionRequiredForSDP and f_ContentType_IsSDP(p_ContentType)) {  /* @sic R5-160794 sic@
                                                                                   NOTE 1: in general a SIP response carries an SDP answer (i.e. no offer) and therefore uses a Require but no Supported header
                                                                                   NOTE 2: when the SDP is optional, ContentType is optional too (-> "ifpresent") and  f_ContentType_IsSDP returns false */
      v_MessageHeader_Response.require := cr_Require(tsc_OptionTagList_precondition);                /* @sic R5-160790, R5-160903, R5-160793 sic@ */
    }
    
    if (not f_IMS_PTC_SecurityScheme_IsGiba() and v_A8) {                       /* @sic R5-130683: pAccessNetworkInfo is optional for GIBA; R5s140092 change 12: CANCEL  sic@
                                                                                   @sic R5-142223: new condition A8: Any response sent by the UE, except for CANCEL requests sic@ */
      v_MessageHeader_Response.pAccessNetworkInfo := v_PAccessNetworkInfo;
    }
    else {
      v_MessageHeader_Response.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;
    }
    if (v_A2A4) {
      if (f_IMS_PTC_AudioFeatureTagIsRequired() ) {                                                                // @sic R5-173021 sic@ @sic R5-177127 sic@
        v_MandatoryContactParams := f_SemicolonParam_List_Add_RX(v_MandatoryContactParams, cr_FeatureParam_Audio); /* @sic R5-165943, R5-170564 sic@
                                                                                                                    * !!!! A11 is for "Response sent by UE", but when the response is not for INVITE or UPDATE there is no Contact header at all !!!! */
      }
      if (v_A18 and v_A20) {                                                                                       /* @sic R5-191117 sic@ */
        v_MandatoryContactParams := f_SemicolonParam_List_Add_RX(v_MandatoryContactParams, cr_FeatureParam_Video);
      }
      if (not v_A5) {
        v_ContactURI := f_Contact_SipUri_HostPortRX(?, v_Port_us);              /* host address needs to be checked after receiving (f_IMS_MessageHeader_CheckContactAddr)
                                                                                   @sic R5s150721, R5-153760: cr_SipUri_HostPort replaced by f_Contact_SipUri_HostPortRX sic@ */

      } else {
        v_ContactURI := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);                /* @sic R5-223454: Contact URI as provided by the UE at call establishment sic@ */
      }

      v_MessageHeader_Response.contact := cr_Contact(v_ContactURI,
                                                     v_MandatoryContactParams); /* @sic R5-165943 sic@ */
    }
  
    if (v_A20) {                        /* response for INVITE
                                           @sic R5-135004 Change 3 sic@
                                           PROSE ISSUE: prose is not clear about whether there shall be any RecordRoute header for responses to requests other than INVITE */
      if (not v_A23) {                  /* @sic R5-223454: condition A23 sic@ */
        v_MessageHeader_Response.recordRoute := p_MessageHeader_Request.recordRoute;     // @sic R5-135004: same as in the INVITE sent to the UE sic@
      }                                 /* @sic R5-223454 - else: RecordRoute remains * sic@ */
    }

    v_MessageHeader_Response.contentType  := p_ContentType;

    return v_MessageHeader_Response;
  }



After Change
	  function f_IMS_OtherResponse_200_MessageHeaderRX(template (value) MessageHeader p_MessageHeader_Request,
                                                   template ContentType p_ContentType := omit,
                                                   boolean p_PreconditionRequiredForSDP := true) runs on IMS_PTC return template (present) MessageHeader
  { /* NOTE: Message is sent by the UE => conditions A2, A4 and A5 are applicable;
     *       For A4 (GIBA) pAccessNetworkInfo is optional */
    /* @sic R5-130683 sic@ */
    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */
    /* @sic R5-160790, R5-160903, R5-160793: p_PreconditionIndication sic@ */
    /* @sic R5-162930: p_PreconditionIndication := PreconditionRequired => per default there is a Require header if the message contains an SDP sic@ */
    /* @sic R5-174544: SessionId removed sic@ */
    /* @sic R5s210615: p_PreconditionsRequiredForSDP (boolean) instead of p_PreconditionIndication (IMS_PreconditionIndication_Type) sic@ */
    var template (present) MessageHeader v_MessageHeader_Response := f_IMS_BaseResponse_MessageHeaderRX(p_MessageHeader_Request);     /* @sic R5s200000: f_IMS_BaseResponse_MessageHeaderRX sic@ */
    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());  /* acc. to A.3.1 condition A5: access network technology and, if applicable, the cell ID */
    var boolean v_A2A4 := false;                                        /* true => response sent by UE for INVITE/UPDATE */
    var boolean v_A5 := f_IMS_PTC_ImsInfo_DialogIsEstablished();        /* true => response sent within a dialog */
    var boolean v_A8 := true;                                           /* true => any response but not CANCEL */
    var boolean v_A20 := false;                                         /* true => response sent by UE for INVITE */
    var boolean v_A18 := pc_IMS_Video_FeatureTag;                       /* true => UE supports video feature tag */
    var boolean v_A23 := false;                                         /* true => response sent for re-INVITE within an established dialog */
    var charstring v_Method := valueof(p_MessageHeader_Request.cSeq.method);
    var integer v_Port_us := f_IMS_PTC_GetPort_us();  /* @sic R5s130985 Change 2 sic@ */
    var template (present) SemicolonParam_List v_MandatoryContactParams := {};
    var template (present) SipUrl v_ContactURI;

    select (v_Method) {
      case ("UPDATE") {
        v_A2A4 := true;
      }
      case ("INVITE") {
        v_A2A4 := true;
        v_A20 := true;
        v_A23 := v_A5;
      }
      case ("CANCEL") {
        v_A8 := false;
      }
    }
    
    if (p_PreconditionRequiredForSDP and f_ContentType_IsSDP(p_ContentType)) {  /* @sic R5-160794 sic@
                                                                                   NOTE 1: in general a SIP response carries an SDP answer (i.e. no offer) and therefore uses a Require but no Supported header
                                                                                   NOTE 2: when the SDP is optional, ContentType is optional too (-> "ifpresent") and  f_ContentType_IsSDP returns false */
      v_MessageHeader_Response.require := cr_Require(tsc_OptionTagList_precondition);                /* @sic R5-160790, R5-160903, R5-160793 sic@ */
    }
    
    if (not f_IMS_PTC_SecurityScheme_IsGiba() and v_A8) {                       /* @sic R5-130683: pAccessNetworkInfo is optional for GIBA; R5s140092 change 12: CANCEL  sic@
                                                                                   @sic R5-142223: new condition A8: Any response sent by the UE, except for CANCEL requests sic@ */
      v_MessageHeader_Response.pAccessNetworkInfo := v_PAccessNetworkInfo;
    }
    else {
      v_MessageHeader_Response.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;
    }
    if (v_A2A4) {
      if (f_IMS_PTC_AudioFeatureTagIsRequired() ) {                                                                // @sic R5-173021 sic@ @sic R5-177127 sic@
        v_MandatoryContactParams := f_SemicolonParam_List_Add_RX(v_MandatoryContactParams, cr_FeatureParam_Audio); /* @sic R5-165943, R5-170564 sic@
                                                                                                                    * !!!! A11 is for "Response sent by UE", but when the response is not for INVITE or UPDATE there is no Contact header at all !!!! */
      }
      if (v_A18 and v_A20) {                                                                                       /* @sic R5-191117 sic@ */
        v_MandatoryContactParams := f_SemicolonParam_List_Add_RX(v_MandatoryContactParams, cr_FeatureParam_Video);
      }
      if (not v_A5) {
        v_ContactURI := f_Contact_SipUri_HostPortRX(?, v_Port_us);              /* host address needs to be checked after receiving (f_IMS_MessageHeader_CheckContactAddr)
                                                                                   @sic R5s150721, R5-153760: cr_SipUri_HostPort replaced by f_Contact_SipUri_HostPortRX sic@ */

      } else {
        v_ContactURI := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);                /* @sic R5-223454: Contact URI as provided by the UE at call establishment sic@ */
       v_ContactURI.urlParameters := *;
      }

      v_MessageHeader_Response.contact := cr_Contact(v_ContactURI,
                                                     v_MandatoryContactParams); /* @sic R5-165943 sic@ */
    }
  
    if (v_A20) {                        /* response for INVITE
                                           @sic R5-135004 Change 3 sic@
                                           PROSE ISSUE: prose is not clear about whether there shall be any RecordRoute header for responses to requests other than INVITE */
      if (not v_A23) {                  /* @sic R5-223454: condition A23 sic@ */
        v_MessageHeader_Response.recordRoute := p_MessageHeader_Request.recordRoute;     // @sic R5-135004: same as in the INVITE sent to the UE sic@
      }                                 /* @sic R5-223454 - else: RecordRoute remains * sic@ */
    }

    v_MessageHeader_Response.contentType  := p_ContentType;

    return v_MessageHeader_Response;
  }



